Chapter 2

graph formating

```
thinn = AbsoluteThickness[.5];
medum = AbsoluteThickness[1.];
thick = AbsoluteThickness[1.5];
black = GrayLevel[0];
BGray = GrayLevel[0.3];
WGray = GrayLevel[0.6];
SetOptions[Plot, PlotStyle → {{thinn, Black}, {thinn, Black}, {thinn, Black}},
  PlotPoints → 40, ImageSize → 360,
  FrameStyle -> medum, AxesStyle -> medum,
  BaseStyle -> {FontFamily -> "Helvetica", FontSlant → Plain, FontSize → 12}];
SetOptions[ListPlot, AxesStyle -> medum, PlotStyle → medum, ImageSize → 384,
  BaseStyle -> {FontFamily -> "Helvetica", FontSlant -> "Plain", FontSize → 12}];
SetOptions[ParametricPlot, PlotStyle →
   {{thinn, Black}, {thinn, Black}, {thinn, Black}}, PlotPoints → 40,
  FrameStyle -> medum, AxesStyle -> medum, PlotStyle → medum,
  BaseStyle -> {FontFamily -> "Helvetica", FontSlant -> "Plain", FontSize → 12}];
SetOptions[Graphics, BaseStyle ->
   {FontFamily -> "Helvetica", FontSlant -> "Plain", FontSize → 12}];
```

2.5 ,2.6 Utility Function to Demand

With the Cobb-Douglas utility form given below as a function of two goods, the water demand function given below will result. Max U s.t. x+pw=income with x and w as decision variables and a,b,&p as parameters.

```
w =.;
a = 0.003;
b = 0.35;
x = 2.;
income = 100.;
u = a * x * w^b;
wd = b * income / (p * (1 + b));
pl25a = Plot[u, {w, 0, 20.},
   PlotRange → {0., 0.022},
   AxesLabel → {"w", "U"},
   Ticks → None,
   AspectRatio → 0.4
  ];
pl25 = Show[pl25a,
  Graphics[Text["U(w)", {12.5, u /. w → 16.}]]
 ]
```




```
pl26a = Plot[25.9259 / w, {w, 2, 20},
   AxesOrigin → {0, 0},
   AxesLabel → {"w", "p"},
   Ticks → {{{10, "w⁰", 0}}, {{2.59259259, "p⁰", 0}}},
   AspectRatio → 0.5
  ];
pl26 = Show[pl26a,
  Graphics[Text["w=D(p)", {4.9, 8.3}]],
  Graphics[{Dashing[{.01, .01}],
    {Line[{{10, 0}, {10, 2.59259259}}], Line[{{0, 2.59259259}, {10, 2.59259259}}]}}]
 ]
```



2.7 Adding Linear Demands

```
{a, b, c, d, e, f} = {10, 2, 7, 0.5, 5, 0.2};
p1 = a - b * w;
p2 = c - d * w;
p3 = e - f * w;
p = Which[0 <= w <= a / b - c / b, .06 + p1,
    a / b - c / b < w < a / b + c / d - e * (1 / b + 1 / d),
    (a / b + c / d - w) / (1 / b + 1 / d),
    a / b + c / d - e * (1 / b + 1 / d) <= w,
    (a / b + c / d + e / f - w) / (1 / b + 1 / d + 1 / f)];
pl27i = Plot[{p1, p2, p3}, {w, 0, 15},
AxesLabel -> {"w", "p"},
  PlotRange -> {0, 8},
  Ticks -> None,
  AspectRatio -> 0.6];
pl27ii = Plot[p, {w, 0, 15},
AxesLabel -> {"w", "p"},
  PlotRange -> {0, 8},
  PlotStyle -> {thick, Black},
  Ticks -> None,
  AspectRatio -> 0.6];
pl27 = Show[pl27i, pl27ii,
   Graphics[{Dashing[{.01, .01}], thinn, Line[{{0, c}, {a / b - c / b, c}}],
     Line[{{0, e}, {a / b + c / d - e * (1 / b + 1 / d), e}}]}],
Graphics[Text["D^1 (p)", {5, 2}]],
Graphics[Text["D^2 (p)", {13.2, 1}]],
Graphics[Text["D^3 (p)", {14, 2.7}]],
Graphics[Text["D(p)", {13, 4.5}]]]
```



2.8 Detour; Benefit Calculations

Linear

```
p =.;
w =.;
q0 = 35 000.;
p0 = 3.;
elast = -0.5;
m = elast * q0 / p0;
b = q0 - m * p0;
wlin = m * p + b
plin = p /. Flatten[Simplify[Solve[w == %, p]]]
```

52 500. - 5833.33 p

9. - 0.000171429 w

LogLinear

```
c = q0 / p0^elast;
wlog = c * p^elast
plog = p /. Flatten[Simplify[Solve[w == %, p]]]
```

$$\frac{60\,621.8}{p^{0.5}}$$

Solve::ifun : Inverse functions are being used by Solve, so
some solutions may not be found; use Reduce for complete solution information. ≫

$$\frac{3.675 \times 10^{9}}{w^{2.000000000000000}}$$

Plot and NB

```
pl28i = Plot[{plin, plog}, {w, 0, q0 * 52 / 35},
   AxesLabel → {w, p},
   PlotRange → {0, 16},
   AspectRatio → 0.618];
pl28 = Show[pl28i,
  Graphics[Text["a", {17 500, 1.5}]],
  Graphics[Text["b", {11 000, 5}]],
  Graphics[Text["c", {11 000, 12}]],
  Graphics[Text["w = -5833p+52500", {10 100, 9.2}]],
  Graphics[Text["w = 60622p^-0.5", {24 300, 13}]],
  Graphics[{Dashing[{.01, .01}], thinn,
    Line[{{35 000, 0}, {35 000, 3}}], Line[{{0, 3}, {35 000, 3}}]}]]
```



Total Benefits

$$\int_0^{q0} \mathbf{plin}\, dw$$

209 996.

$$\int_0^{q0} \mathbf{plog}\, dw$$

Integrate::idiv : Integral of $\frac{1}{w^{2.000000000000000}}$ does not converge on {0, 35000.}. ≫

$$\int_0^{35\,000.} \frac{3.675 \times 10^9}{w^{2.000000000000000}}\, dw$$

Gross Loss

$$\int_{0.8*q0}^{q0} \mathbf{plin}\, dw$$

25 198.4

$$\int_{0.8*q0}^{q0} \text{plog}\, dw$$

26 248.3

Net Loss

$$\int_{0.75*q0}^{q0} (\text{plin} - 3)\, dw$$

6562.16

$$\int_{0.75*q0}^{q0} (\text{plog} - 3)\, dw$$

8749.46

2.9 MNB

```
mc = 1.95 + 0.00003 * w;
mnblog = plog - mc
pl29i = Plot[{mc, plog, mnblog}, {w, 0, q0 * 52 / 35},
    AxesLabel → {w, p},
    PlotRange → {0, 16},
    AspectRatio → 0.5];
```

$$-1.95 + \frac{3.675\times 10^{9}}{w^{2.000000000000000}} - 0.00003\, w$$

```
pl29 = Show[pl29i,
  Graphics[Text["MNB", {17 350, 6.5}]],
  Graphics[Text["MB", {21 400, 10}]],
  Graphics[Text["MC", {48 000, 4}]],
  Graphics[{Dashing[{.01, .01}], thinn,
    Line[{{35 000, 0}, {35 000, 3}}], Line[{{0, 3}, {35 000, 3}}]}]]
```



2.10, 2.11 Two MNBs

```
plog2 = plog * 0.5;
mc2 = 1.5 + 0.00002 * w;
mnblog2 = plog2 - mc2;
flipmnblog2 = mnblog2 /. w → (50 000 - w);
FindRoot[mnblog - flipmnblog2 == 0, {w, 25 000}]
mnblog /. %
```

```
{w → 28 056.2}
```

```
1.87707
```

optimal w for 2nd sector?

```
50 000 - 28 056.16
```

```
21 943.8
```

MC at optimal w's for both sectors?

```
mc /. w → 28 056.2
mc2 /. w → 21 943.84
```

```
2.79169
```

```
1.93888
```

```
pl211a = Plot[{mnblog, flipmnblog2}, {w, 0, q0 * 48.8 / 35},
    Frame → {{True, True}, {True, False}},
    FrameLabel → {"w", "", "", ""},
    PlotRange → {0, 16},
    Frame → True,
    AspectRatio → 0.6];
pl211 = Show[pl211a,
  Graphics[Text["MNB₁", {19 300, 13}]],
  Graphics[Text["MNB₂", {35 600, 13}]],
  Graphics[{Dashing[{.01, .01}], thinn, Line[{{0, 1.877}, {50 000, 1.877}}]}],
  ImageSize → 360]
```



I find w's on each curve corresponding to the same p, and then add them. I create a table of such points spanning the interesting p range. Then I connect to dots in a plot.

```
points1 = Table[{w /. FindRoot[mnblog == p, {w, 20 000}], p / 2}, {p, 0, 15.8, .2}];
points2 = Table[{w /. FindRoot[mnblog2 == p, {w, 20 000}], p / 2}, {p, 0, 15.8, .2}];
points = points1 + points2;
pl210i =
  ListPlot[points, AxesLabel → {"w", "p"}, Joined → True, PlotRange → {0, 16},
   AxesOrigin → {0, 0},
   AspectRatio → 0.6];
pl210ii =
  Plot[{mnblog, mnblog2}, {w, 0, 65 000}, AxesLabel → {"w", "p"}, PlotRange → {0, 16},
   AspectRatio → 0.6];
pl210 = Show[{pl210i, pl210ii},
  Graphics[Text["MNB₁", {19 400, 13}]],
  Graphics[Text["MNB₂", {8600, 11}]],
  Graphics[Text["MNB₁ + MNB₂", {38 300, 10}]],
  Graphics[{Dashing[{.01, .01}], thinn,
    Line[{{21 944, 0}, {21 944, 1.87707}}], Line[{{28 056, 0}, {28 056, 1.87707}}],
    Line[{{50 000, 0}, {50 000, 1.87707}}], Line[{{0, 1.87707}, {50 000, 1.87707}}]}],
  ImageSize → 360
 ]
```



2.12 Two Sectors with One Yielding Return Flow

Method 1 - graphed

```
R = 0.25;
flipmnblog2R = mnblog2 /. w → (50 000 - (1 - R) * w)
```

$$-1.5 + \frac{1.8375 \times 10^{9}}{(50\,000 - 0.75\,w)^{2.000000000000000}} - 0.00002\ (50\,000 - 0.75\,w)$$

```
FindRoot[mnblog / (1 - R) - flipmnblog2R == 0, {w, 25 000}]
mnblog /. %
```

{w → 32 407.3}

0.576995

Method 2 - closer to opt. model

```
mnb1 = mnblog /. w → w1
mnb2 = mnblog2 /. w → w2
```

$$-1.95 + \frac{3.675 \times 10^9}{w_1^{2.000000000000000}} - 0.00003\, w_1$$

$$-1.5 + \frac{1.8375 \times 10^9}{w_2^{2.000000000000000}} - 0.00002\, w_2$$

```
FindRoot[{mnb1 == (1 - R) * mnb2, (1. - R) * w1 + w2 == 50 000}, {w1, 25 000}, {w2, 25 000}]
```

$\{w_1 \to 32\,407.3,\ w_2 \to 25\,694.5\}$

```
{mnb1, mnb2} /. %
```

{0.576995, 0.769327}

plot

```
pl212a = Plot[{mnblog, mnblog / (1 - R), flipmnblog2R}, {w, 0, q0 * 48.8 / 35},
    Frame → {{True, True}, {True, False}},
    FrameLabel → {"w", "", "", ""},
    PlotRange → {0, 2.8},
    Frame → True,
    PlotStyle → {{Dashing[{0.01, 0.01}], {black, thinn}},
      {Dashing[{1, 0}], {black, thinn}}, {Dashing[{1, 0}], {black, thinn}}},
    AspectRatio → 0.6];
pl212 = Show[pl212a,
  Graphics[Text["MNB1", {24 900, 1.7}]],
  Graphics[Text["MNB1/(1 - R')", {32 000, 2.3}]],
  Graphics[Text["MNB2", {43 000, 2.4}]],
  Graphics[{Dashing[{.01, .01}], thinn,
    Line[{{32 407, 0.}, {32 407, 0.77}}],
    Line[{{-400, 0.577}, {32 407, 0.577}}],
    Line[{{32 407, 0.77}, {49 500, 0.77}}]}],
  ImageSize → 400]
```